
Abstract. We introduce a simple approach extending the input language of An-
swer Set Programming (ASP) systems by multi-valued propositions. Our ap-
proach is implemented as a (prototypical) preprocessor translating logic programs
with multi-valued propositions into logic programs with Boolean propositions
only. Our translation is modular and heavily benefits from the expressive input
language of ASP. The resulting approach, along with its implementation, allows
for solving interesting constraint satisfaction problems in ASP, showing a good
performance.

1 Introduction

Boolean constraint solving technologies like Satisfiability Checking (SAT;[1]) and An-
swer Set Programming (ASP;[2]) have demonstrated their efficiency and robustness
in many real-world applications, like planning [3, 4], model checking [5, 6], and bio-
informatics [7, 8]. However, many applications are more naturally modeled by addition-
ally using non-Boolean propositions, like resources or functions over finite domains.
Unlike in SAT, however, where such language extensions are application-specific, ASP
offers a rich application-independent modeling language. The high level of expressive-
ness allows for an easy integration of new language constructs, as demonstrated in the
past by preferences [9] or aggregates [10]. Interesting examples of language extensions
illustrating the utility of mixing Boolean and non-Boolean propositions can be found
in [11–13], dealing with reasoning about actions.

In fact, a Boolean framework seems to offer such an elevated degree of efficiency
that it becomes also increasingly attractive as a target language for non-Boolean con-
straint languages. This is for instance witnessed by the system Sugar [14], an award-
winning SAT-based constraint solver. This motivated us to pursue a translational ap-
proach rather than an integrative one, as proposed in [15, 16] or, in more generality, in
the field of SAT modulo theories.

In what follows, we expect the reader to be familiar with ASP (cf. [2]) as well as
the input language of lparse [17, 18] or gringo [19, 20].

2 Approach

Our approach takes a logic program with multi-valued propositions and translates it into
a standard logic program having Boolean propositions only. A multi-valued proposition
is a variable taking exactly one value out of a pre-defined range. Currently, the range
of multi-valued propositions is fixed to integer intervals. A multi-valued proposition is
subject to two functional conditions, namely, it takes at most and at least one value. This
is conveniently expressed by means of cardinality constraints. Let us make this precise
by looking at the current input syntax.

For instance, the multi-valued proposition v taking values between 1 and 3 is de-
clared as follows:

#variables v = 1..3.

Such a declaration is translated into the following expressions:

x dom(v, 0, 1..3).
1 {val(v,Val) : x dom(v, 0,Val)} 1.

(1)

The new ternary predicate x dom captures the fact that v has arity 0 and ranges over
1 to 3. This predicate is hidden in the output via #hide x dom(X ,Y ,Z ). The value
assignment to v is captured by the binary predicate val . Thus passing the logic program
in (1) to an ASP system yields three answer sets, given by {val(v, 1)}, {val(v, 2)}, and
{val(v, 3)}, each representing a valid assignment to variable v.

The declaration #variables also allows for more fine-grained specifications, like:

#variables u, v = 1..3 | 10..20.

#variablesf(X) = 1..10 :- p(X).

The first declaration shows how multiple variables can be specified, sharing a non-
consecutive range of values. The second one shows how terms can be incorporated. For
this, a term’s domain must be guarded by domain predicates, like p(X). These domain
predicates are then added as body literals to the resulting cardinality constraint, viz.:

1 {val(f(X),Val) : x dom(f, 1,Val)} 1 :- p(X).

Interestingly, the mere possibility of defining multi-valued propositions opens up
the possibility of specifying and solving simple constraint satisfaction problems. As
an example, take two variables u, v ranging over {1, 2, 3} and being subject to the
constraint u + v ≤ 3. This can be expressed by means of the following program:

#variables u, v = 1..3.

u + v ≤ 3.

This program is then translated as follows. While the declaration of u and v is given as
in (1), the constraint u + v ≤ 3 is expressed via an integrity constraint:

:- val(u,Valu), val(v,Valv ),Valu + Valv > 3.

Note that the original constraint u + v ≤ 3 appears negated as u + v > 3 within the
integrity constraint. The idea is to exclude assignments to u and v such that u + v > 3.
Finally, we note that passing the result of the compilation to an ASP system yields three
answer sets, containing

{val(u, 1), val(v, 1)}, {val(u, 1), val(v, 2)}, and {val(u, 2), val(v, 1)}.
The above transformation applies whenever an arithmetic expression involving

multi-valued propositions appears in the head of a rule. Appearances of such expres-
sions as body literals can be dealt with in a similar way. Unlike above, however, the
original constraint is not negated. For instance, the rule within the program

#variables u, v = 1..3.

p(u, v) :- u + v ≤ 3.

is turned into

p(Valu ,Valv ) :- val(u,Valu), val(v,Valv ),Valu + Valv ≤ 3.

The resulting program has nine answer sets, reflecting all possible value assignments
to u and v. However, among them, only three contain a single instance of predicate p,
namely, p(1, 1), p(1, 2), and p(2, 1).

Finally, our approach provides a dedicated treatment of the popular alldistinct con-
straint, expressing that all involved variables must take pairwisely different values. As
before, this constraint is easily mapped onto cardinality constraints. To illustrate this,
consider the program:

#variables u, v, w = 1..3.

#alldistinct u, v, w.

The declaration of u, v, and w is dealt with as in (1). The alldistinct constraint yields
the following program rules:

#hide x distinct 0 var(Var). #hide x distinct 0 val(Val).
x distinct 0 var(u). x distinct 0 val(Val) :- x dom(u, 0,Val).
x distinct 0 var(v). x distinct 0 val(Val) :- x dom(v, 0,Val).
x distinct 0 var(w). x distinct 0 val(Val) :- x dom(w, 0,Val).
:- x distinct 0 val(Val), 2 {val(Var ,Val) : x distinct 0 var(Var)}.

The predicates x distinct 0 var and x distinct 0 val are unique for each
alldistinct constraint, fixing the sets of involved variables and values, respectively. The
integrity constraint is violated whenever there are at least two variables sharing a value.

Our translation tool xpanda is written in Python and best used via Unix’ pipes, e.g.:

cat simple.lp | xpanda.py | gringo | clasp 0.

A prototype version implementing a subset of the above transformations is (presently)
available at http://files.mutaphysis.de/xpanda.zip. It works with ASP systems support-
ing the input language of lparse [17, 18] or gringo [19, 20]. Clearly, the translation of
xpanda can easily be modified to using disjunction and explicit counting aggregates
rather than cardinality constraints, and then be used by ASP systems like dlv [21].

3 A (Little) Case Study: SEND+MORE=MONEY

Let us conduct a brief case-study reflecting the scalability of our approach. To this end,
we consider the SEND+MORE=MONEY puzzle. The task is to assign distinct numbers
from {0, . . . , 9} to the variables S, E, N, D, M, O, R, Y such that the addition of the
decimal numbers SEND and MORE results in the decimal number MONEY . By
convention, leading digits of decimal numbers must not be 0. This eliminates 0 from
the domains of S and M . Moreover, we know that M cannot be greater than 1 because
it occurs as carry. Hence, the value of M must be 1, effectively reducing the variables
to S, E, N, D, O, R, Y . For clarity, however, we below use variable notation for M too.

A first and apparently compact representation of this problem is the following one:

#variables m = 1.
#variables s = 2..9.
#variables e,n,d,o,r,y = 2..9 | 0.
#alldistinct s,e,n,d,o,r,y.

s*1000+e*100+n*10+d
+ m*1000+o*100+r*10+e

== m*10000+o*1000+n*100+e*10+y.

The result of the compilation is given in Appendix A. Unfortunately, the grounding
blows up in space because the (non-ground) integrity constraint resulting from the ac-
tual SEND+MORE=MONEY constraint leads to 8 ∗ 96 ground integrity constraints.

This extreme blow-up is avoided in the following representation, using column-wise
addition and three carry variables to express the SEND+MORE=MONEY constraint:

#variables m = 1.
#variables s = 2..9.
#variables e,n,d,o,r,y = 2..9 | 0.
#alldistinct s,e,n,d,o,r,y.
#variables n1,e1,y1 = 0..1.

d+e == y+y1*10.
n+r+y1 == e+e1*10.
e+o+e1 == n+n1*10.
s+m+n1 == o+ m*10.

The result of the compilation is given in Appendix B. Unlike a single constraint with
seven variables, this formalization relies on four constraints with at most five variables.
This reduces the resulting ground program to 7172 rules, which the ASP solver clasp
(1.2.1) solves in milliseconds. The overall runtime, including xpanda, gringo (2.0.3),
and clasp, is less than half a second when enumerating all solutions. In fact, this exam-
ple has a unique solution containing:

val(s,9) val(e,5) val(n,6) val(d,7)
val(m,1) val(o,0) val(r,8)

val(n1,0) val(e1,1) val(y1,1) val(y,2).

4 Conclusion

We have provided a simple transformation-based approach to incorporating multi-
valued propositions into ASP. Our translation is modular and heavily benefits from the
expressive input language of ASP, providing variables and aggregate statements such
as cardinality constraints. Once multi-valued propositions are available, it is possible to
formulate and solve interesting constraint satisfaction problems by appeal to ASP tech-
nology. As with many ASP applications, the bottleneck of the approach manifests itself
in grounding. We have seen that constraints involving too many variables may result in
a space blow-up. This phenomenon can to some extent be controlled by the user since
the number of variables remains the same in the initial specification and the resulting
compilation. Of course, large domains may still be problematic.

Many open questions remain, concerning encoding optimizations, further language
constructs, etc., and are subject to future research.

Acknowledgments. We are grateful to Wolfgang Faber for commenting on this paper.
This work was partially funded by DFG under Grant SCHA 550/8-1 and by the Go-
FORSYS1 project under Grant 0313924.

A First SEND+MORE=MONEY Representation: Compilation

:- val(s,Val_s), val(e,Val_e), val(n,Val_n), val(d,Val_d),
val(m,Val_m), val(o,Val_o), val(r,Val_r), val(y,Val_y),

Val_s*1000+Val_e*100+Val_n*10+Val_d
+ Val_m*1000+Val_o*100+Val_r*10+Val_e

!= Val_m*10000+Val_o*1000+Val_n*100+Val_e*10+Val_y.

#hide _x_distinct_0_var(X).
#hide _x_distinct_0_val(X).

_x_distinct_0_var(s). _x_distinct_0_val(Val) :- _x_dom(s,0,Val).
_x_distinct_0_var(e). _x_distinct_0_val(Val) :- _x_dom(e,0,Val).
_x_distinct_0_var(n). _x_distinct_0_val(Val) :- _x_dom(n,0,Val).
_x_distinct_0_var(d). _x_distinct_0_val(Val) :- _x_dom(d,0,Val).
_x_distinct_0_var(o). _x_distinct_0_val(Val) :- _x_dom(o,0,Val).
_x_distinct_0_var(r). _x_distinct_0_val(Val) :- _x_dom(r,0,Val).
_x_distinct_0_var(y). _x_distinct_0_val(Val) :- _x_dom(y,0,Val).

:- _x_distinct_0_val(Val),
2{ val(Var,Val) : _x_distinct_0_var(Var) }.

#hide _x_dom(X,Y,Z).

_x_dom(m,0,1).
1{ val(m,X_D_Val) : _x_dom(m,0,X_D_Val) }1.

1 http://www.goforsys.org

_x_dom(s,0,2..9).
1{ val(s,X_D_Val) : _x_dom(s,0,X_D_Val) }1.

_x_dom(e,0,2..9). _x_dom(e,0,0).
1{ val(e,X_D_Val) : _x_dom(e,0,X_D_Val) }1.

_x_dom(n,0,2..9). _x_dom(n,0,0).
1{ val(n,X_D_Val) : _x_dom(n,0,X_D_Val) }1.

_x_dom(d,0,2..9). _x_dom(d,0,0).
1{ val(d,X_D_Val) : _x_dom(d,0,X_D_Val) }1.

_x_dom(o,0,2..9). _x_dom(o,0,0).
1{ val(o,X_D_Val) : _x_dom(o,0,X_D_Val) }1.

_x_dom(r,0,2..9). _x_dom(r,0,0).
1{ val(r,X_D_Val) : _x_dom(r,0,X_D_Val) }1.

_x_dom(y,0,2..9). _x_dom(y,0,0).
1{ val(y,X_D_Val) : _x_dom(y,0,X_D_Val) }1.

B Second SEND+MORE=MONEY Representation: Compilation

:- val(d,Val_d), val(e,Val_e), val(y,Val_y), val(y1,Val_y1),
Val_d+Val_e != Val_y+Val_y1*10.

:- val(n,Val_n), val(r,Val_r), val(y1,Val_y1), val(e,Val_e),
val(e1,Val_e1), Val_n+Val_r+Val_y1 != Val_e+Val_e1*10.

:- val(e,Val_e), val(o,Val_o), val(e1,Val_e1), val(n,Val_n),
val(n1,Val_n1), Val_e+Val_o+Val_e1 != Val_n+Val_n1*10.

:- val(s,Val_s), val(m,Val_m), val(n1,Val_n1), val(o,Val_o),
Val_s+Val_m+Val_n1 != Val_o+Val_m*10.

#hide _x_distinct_0_var(X).
#hide _x_distinct_0_val(X).

_x_distinct_0_var(s). _x_distinct_0_val(Val) :- _x_dom(s,0,Val).
_x_distinct_0_var(e). _x_distinct_0_val(Val) :- _x_dom(e,0,Val).
_x_distinct_0_var(n). _x_distinct_0_val(Val) :- _x_dom(n,0,Val).
_x_distinct_0_var(d). _x_distinct_0_val(Val) :- _x_dom(d,0,Val).
_x_distinct_0_var(o). _x_distinct_0_val(Val) :- _x_dom(o,0,Val).
_x_distinct_0_var(r). _x_distinct_0_val(Val) :- _x_dom(r,0,Val).
_x_distinct_0_var(y). _x_distinct_0_val(Val) :- _x_dom(y,0,Val).

:- _x_distinct_0_val(Val),
2{ val(Var,Val) : _x_distinct_0_var(Var) }.

#hide _x_dom(X,Y,Z).

_x_dom(m,0,1).

1{ val(m,X_D_Val) : _x_dom(m,0,X_D_Val) }1.

_x_dom(s,0,2..9).
1{ val(s,X_D_Val) : _x_dom(s,0,X_D_Val) }1.

_x_dom(e,0,2..9). _x_dom(e,0,0).
1{ val(e,X_D_Val) : _x_dom(e,0,X_D_Val) }1.

_x_dom(n,0,2..9). _x_dom(n,0,0).
1{ val(n,X_D_Val) : _x_dom(n,0,X_D_Val) }1.

_x_dom(d,0,2..9). _x_dom(d,0,0).
1{ val(d,X_D_Val) : _x_dom(d,0,X_D_Val) }1.

_x_dom(o,0,2..9). _x_dom(o,0,0).
1{ val(o,X_D_Val) : _x_dom(o,0,X_D_Val) }1.

_x_dom(r,0,2..9). _x_dom(r,0,0).
1{ val(r,X_D_Val) : _x_dom(r,0,X_D_Val) }1.

_x_dom(y,0,2..9). _x_dom(y,0,0).
1{ val(y,X_D_Val) : _x_dom(y,0,X_D_Val) }1.

_x_dom(n1,0,0..1).
1{ val(n1,X_D_Val) : _x_dom(n1,0,X_D_Val) }1.

_x_dom(e1,0,0..1).
1{ val(e1,X_D_Val) : _x_dom(e1,0,X_D_Val) }1.

_x_dom(y1,0,0..1).
1{ val(y1,X_D_Val) : _x_dom(y1,0,X_D_Val) }1.

